

Operational Uses of Patient Profiles ... Even in an eCTD and SDTM World

Terek Peterson, Octagon Research Solutions, Wayne, PA
Sanjiv Ramalingam, Octagon Research Solutions, Wayne, PA

ABSTRACT

For years, patient profiles were instrumental in conveying information and data about a single patient in a concise output within a company's submission to FDA. In the clinical trial world of eCTD and SDTM, the need for these outputs has decreased because when data submitted is SDTM compliant the FDA has a powerful graphical patient profile tool to review the data. What is now a benefit of FDA receiving SDTM domains leaves other stakeholders within the sponsor company creating the submission without outputs that many groups find essential to do their work. Through experience, there is still a need for the creation of simple patient profiles to meet operational objectives when conducting clinical trials, creating the submission documents, and making patient safety decisions.

The techniques the authors of this paper are proposing are for operational uses of patient profiles as in narrative writing, Data Safety Monitoring Board (DSMB), and internal data review meetings. A highly efficient and robust algorithm in implementing a patient profile will be discussed in the paper. Numerous features of ODS RTF in conjunction with PROC REPORT and PROC PRINT have been used to make the presentation of the report very robust. This paper will not provide the techniques for bookmarking or hyperlinking profile PDFs, but know that the more current versions of SAS or the use of ODS PS have this capability.

INTRODUCTION

The patient profile is made up of small macro programs or modules each with a unique purpose. Multiple modules can print to a single page, allowing a mix of procedures on one page of output. The macro automatically creates profiles for a single patient or multiple patients depending on the macro call. The implementation is very structured i.e. all data manipulations required to create datasets to be used in PROC REPORT are implemented separately. A separate looping is then used to create the reports for individual patients.

Part of this technique is to subset all datasets within a library of data by a single patient number. If particular datasets, like laboratory data, are very large a trick was found to make subfolders within the WORK directory. The advantage to this technique is to read in the entire permanent library of data into WORK once, so that the subsetting routines for each patient are more efficient. These subfolders can be automatically deleted easily once a patient's profile is created. This avoids the contamination of data between patient profiles.

Now that all the data for one patient has been subsetted, many different SAS procedures can be used together to present the data. It depends on the requirement of the customer as to how data will be displayed. PROC PRINT profiles offer a quick technique for data dumps for medical writers. PROC REPORT and graphic procedures might be used if the requirements of the customer specify a more structured layout.

Of note, this patient profile technique could be used in a submission if the requirements laid out in the FDA's study data specifications document are followed. The document states, "Each individual patient's complete patient profile is in a single PDF file. Including the patient ID in the file name will help identify the file. Alternatively, all patient profiles for an entire study may be in one file if the size of each individual patient profile is small and there are not a large number of patient profiles needed for the study. If you do the latter, bookmark the PDF file using the patient's ID. Including the study number in the file name will help identify the file."

Also, it is the experience of the authors that ODS PS, instead of ODS PDF, offers the most flexibility for customizing outputs if special requirements are needed. Postscript output must be then rendered which creates an additional step in the profile creation process.

ABBREVIATIONS

CRT	Case Report Tabulation	DSMB	Data Safety Monitoring Board
FDA	Food and Drug Administration	CSR	Clinical Study Report
PDF	Portable Document Format	PS	Post Script

USE OF PATIENT PROFILES

With the submission of clinical datasets in SDTM as part of an eCTD, the FDA is now equipped with tools to automatically generate listings and patient profiles of patient data; hence the FDA does not necessarily require that the listings or patient profiles be generated unless requested or applicable. Through a cooperative research and development agreement (CRADA), CSS Informatics and the FDA created an impressive graphical display of patient information. PPD Patient Profiles, that are available to FDA reviewers, are more powerful than reviewing data in a tabular output. But, what about the rest of us who need patient profiles and have limited budgets?

Even though the FDA currently does not require patient profiles, there are other stages of the clinical trial process that require the use of patient profiles namely DSMB meetings, submission narrative writing, dose escalation meetings, and pre-database lock data quality reviews. The example in the following section shows a simple patient profile that could be used to give medical writers an early look at the data before listings, figures, and tables are produced from the analysis datasets. This allows the medical writers to, for example, create the patient narratives before working on the clinical study report. The second example on the next page is more structured and could be used for a DSMB meeting where a dump of the data might be insufficient due to the audience.

USING PROC PRINT FOR THE BODY

This example uses simple PROC PRINT statements to produce the body of the profile. With the use of the ODS STARTPAGE=NEVER option, multiple outputs from any SAS procedure can be displayed on one page.

Depending on the requirements, the use of PROC PRINT has advantages and disadvantages. The advantage is that the SAS code is pretty simple to create and there aren't many options. The disadvantage is also there aren't many options and the layout of output is determined by the SAS system when the patient profile is created. If the customer of the profile needs the same layout per patient, PROC REPORT offers more control and options.

The ODS escape character with options can be used to also format specific data points. One example of possible use can be seen below. This patient has a serious adverse event of Pancreatitis so a patient narrative should be created for the submission. It will be shown later in the paper that ODS allows for the **bolding** or **coloring** of data points to highlight items of interest.

Bogusene		Patient Profile						Page 1 of 3		
Patient: BOG-030452 Treatment: BOGUSENE-XR Gender/Age/Race: F/58/CAUCASIAN Baseline Hgt (cm): 162 Wgt (kg): 74										
Completed: YES Discontinued: NO Reason: COMPLETED Date of Study Disc/Completion/Lost to Followup: 10JUL1998										
Schedule: DOSE STRT LAST DOSE STUDY END										
22MAY1996 18JUN1998 10JUL1998										
CR	Study Identifier	Study Site Identifier	Subject Identifier for the Study	Subject Reference Start Date/Time	Subject Reference End Date/Time	Date/Time of Birth	Age in AGEU at Reference Date/Time	Age Units	Sex	Race
1	0123-321US	123	321	1996-05-22	1998-06-18	1937-07-08	58	YEARS	F	WHITE
DR	Ethnicity	Description of Arm	Planned Arm Code	Country						
1		BOGX	BOGUSENE-XR	USA						
CR	Study Identifier	Reported Name of Drug, Med, or Therapy	Dose/Units	Dosing Frequency Per Interval	Route of Administration	Start Date/Time of Medication	End Date/Time of Medication	Ongoing?		
1	0123-321US	BACLOFEN	40 MG	TID PRN	PO		1996-05			
2	0123-321US	DITROPAN	25 MG	QHS PRN	PO		1997-10-15			
3	0123-321US	ELAVIL	5 MG	QHS PRN	PO				DURING/AFTER	
4	0123-321US	BACLOFEN	40 MG	TID PRN	PO	1996-05-22			DURING/AFTER	
5	0123-321US	BIAxin	750 MG	BID PRN	PO	1997-01-03	1997-01-09			
AE	Study Identifier	Reported Term for the Adverse Event	Dictionary-Derived/ Term	Severity/ Serious Intensity	Serious Event					
1	0123-321US	TINGLING PERIORAL	Paraesthesia oral	MILD	N					
2	0123-321US	PANCREATITIS	Pancreatitis	SEVERE	Y					
3	0123-321US	CLONUS	Clonus	MODERATE	N					
4	0123-321US	PSORIATIC LESION R FOREARM + L LEG	Psoriasis	MILD	N					
Programmer: TJP Table Generated: 26JUL2008 12:16										

USING PROC REPORT FOR THE BODY

When exacting layouts are needed in the output, PROC REPORT offers far more control over the display of data. PROC REPORT in combination with ODS offers a plethora of options to customize the output. As can be seen below, options exist to create subheaders and subtitles in any part of the report. The underscore option is also used as part of the column definition in PROC REPORT to underline the topic in each report. Options also exist to present the report with or without gridlines through the use of PROC TEMPLATE.

<u>DEMOGRAPHIC INFORMATION</u>		<u>Patient Profile</u>		Page 1 of 26
Age (years)	Gender	Race	Ethnicity	
37	M	WHITE	NOT HISPANIC OR LATINO	
<u>DISEASE HISTORY</u>				
Subject's Disease History				
THE DISEASE HAS PERSISTED DESPITE SURGERY (FAILED SURGERY)				
ELEVATED PH (ABOVE THE LIMIT OF NORMAL) ON AT LEAST TWO 24-HOUR URINE COLLECTIONS WITHIN 4 MONTHS OF BASELINE (DAY 1)				
<u>MEDICAL HISTORY</u>				
Category	Reported Term	Month/Year of Onset	Ongoing (YES)	
ABDOMINAL	BLOATING	DEC1994	Y	
	OBESE	JUN2008	Y	
CARDIOVASCULAR	HYPERTENSION	---1999	Y	
EXTREMITIES/ MUSCULOSKELETAL	(R) FOOT FRACTURE/ STRESS	---2004	Y	
	(R) WRIST FRACTURE	---2003	Y	
	(L) FOOT FRACTURE/ STRESS	---2002	Y	
Programmer: SR Table Generated: 17DEC2009 16:37				

SUMMARY OF MACROS

The programming technique primarily involves the use of three macros. Each macro has a specific purpose. PROC REPORT will be used to create the reports. Two library sources are used.

One library source, RAW, will contain information for all patients to be used in PROC REPORT. The other library source namely WORK will be used to subset the datasets in the RAW library only for the patient for which the patient profile is being created.

In brief, the order of invoking macros is as follows. The macro %SER_DISC invokes the macro %PROFILE to select the set of patients for a particular domain that, for example, discontinued the study due to an AE or had a serious AE. The macro %PROFILE sets up the ODS template to use and also setups the headers that will appear on each page of the profile. The macro %PP_BODY is then invoked within the macro %PROFILE to organize the subsetting of patient information without having to create permanent datasets.

%SER_DISC – CREATE THE HITLIST

The macro %SER_DISC is responsible for creating macro variables for each patient that later invoke macros to actually create the patient profile. It is assumed that all demographic information is present in the DM dataset is contained in the RAW library. The patient information is contained in the DM dataset under the variable name, SUBJID. A macro variable is created for each of the patients in the DM dataset and the macro %PROFILE is then invoked for the total number of patients using a DO loop.

```

%macro ser_disc();
  proc sql noprint ;
    select distinct subjid into:pat_list1-: pat_list1000 from sdtm.dm;
    select count(distinct subjid) into :mptcnt from sdtm.dm;
  quit;
  %do mm=1 %to &mptcnt. ;
    %profile(subjid=&&pat_list&mm.);
  %end;
%mend ser_disc;

```

%PROFILE – CREATE THE TEMPLATE AND HEADER

The macro %PROFILE contains the template definition to be used in the report and controls the information that will be output in the patient profile. The macro %PROFILE invokes the %PP_BODY macro. Datasets are subset from RAW library to contain information for only the patient under consideration.

```

%macro profile(usubjid=);

  proc template;
  ...
  ..
  .
  end;
  run;
  title1 ;
  title2 ;
  footnote1 ;
  footnote2 ;
  %pp_body();

%mend profile;

```

%PP_BODY – CREATE THE BODY OF THE PROFILE

As can be seen below, the %PP_BODY macro contains the PROC REPORT definition for each of the subsections contained in the patient profile. This example shows the display of information from the SDTM DM and DC, Disease Characteristics, domains. Each SDTM domain would have a PROC REPORT step to customize the display.

```

%macro pp_body;

  %if &dsn=DM %then %do;
    proc report data=DM nowindows headline headskip split="\ " ;
      columns age sex race ethnic;
      define age/display "Age (years)" f=$20. style={cellwidth=1% just=c};
      define sex/display "Gender" f=$20. style={cellwidth=1% just=c};
      define race/display "Race" f=$30. style={cellwidth=1% just=c};
      define ethnic/group "Ethnicity" f=$30. style={cellwidth=1% just=c};
    run;
  %end;

  %if &dsn=DC %then %do;

    proc report data=DC nowindows headline headskip split="\ ";
      column ("~R'\ql'~R'\b'~R'\ul'DISEASE HISTORY~R'\ul0'~R' \b0'"( dterm1 ));
      define trm1/order "Patient's Disease History" style={cellwidth=1%just=1};
    run;
  %end;

%mend pp_body;

```

When using PROC PRINT the macro %PP_BODY would be coded like the example below. The OBS statement can be used to put a brief title for the particular domain that is displayed. Note the use of the ~S to make the text *italic* and **bold**. The WIDTH=MIN option will use the minimum amount of page real estate, so more data can be presented on a single page. Each label statement in the PROC PRINT is somewhat analogous to the define statement in the PROC REPORT by helping to control the width of a column.

```

%macro pp_body;

  %if &nobs>0 %then %do;
    proc print data=&dsn width=min round split=""
              %if &label=Y %then label;
    obs="~S={font_style=italic font_weight=bold font_size=8pt}&dsn~S={}~n~n";

  %if &dsn=DM %then %do;
    var studyid siteid subjid rfstdtc rfendtc brthdte age ageu sex race
        ethnic armcd arm country
    / style(data)={just=left} style(header)={just=left};
    label studyid="Study\Identifier"
           subjid="Patient\Identifier\for the Study"
           rfstdtc="Patient\Reference\Start Date/Time"
           rfendtc="Patient\Reference\End Date/Time"
           siteid="Study\Site\Identifier"
           brthdte="Date/Time\of Birth"
           age="Age in AGEU\at Reference\Date/Time"
           ageu="Age \Units"
           sex="Sex"
           race="Race"
           ethnic="Ethnicity"
           armcd="Description\ of Arm"
           arm="Planned \Arm Code"
           country="Country";

  %end;

  %if &dsn=AE %then %do;
    var studyid aeterm aedecod aesev aeser aeacn aere1 aeout
        aeconrt aestdte aeendte
    / style(data)={just=left} style(header)={just=left};
    label studyid="Study\Identifier"
           aeseq="Sequence\Number"
           aespide="Sponsor-\Defined\Identifier"
           aeterm="Reported Term\for the\Adverse Event"
           aedecod="Dictionary-Derived/ Term"
           aesev="Severity/\Intensity"
           aeser="Serious\Event"
           aeacn="Action Taken \with\ Study\ Treatment"
           aere1="Relationship\to Non-Study\Treatment"
           aeout="Outcome of\Adverse Event"
           aeconrt="Concom-\itant\Trtmnt\Given"
           aestdte="Start Date/Time\of Adverse Event"
           aeendte="End Date/Time\of Adverse Event";

  %end;

  %if &dsn=CM %then %do;

  ETC...

%mend pp_body;

```

TECHNICAL CONSIDERATION

Some new methodologies were followed when creating patient profiles. One of the two areas where these methodologies were implemented involved creating a subfolder within the work folder. The other involves a few tricks using ODS.

CREATING A SUBFOLDER WITHIN THE WORK FOLDER

The objective was to have a dedicated workspace that exclusively contained datasets used in the final reporting stage. The added advantage with this approach is that this workspace would be automatically deleted as soon as the SAS session expired and it required no explicit code to clear this work area.

Creating the subfolder within the work folder involves four easy steps. In the first step the address of the work folder is obtained using the system function, SYSFUNC. In the second step a macro variable that contains the address of the current work folder in addition to the user created subfolder name is created. This is implemented by concatenating the current work folder address as obtained in step one to a string that contains the user preferred name of the subfolder. In the third step, two DOS commands are implemented using the 'X' statement. The location of the current work directory is addressed by using the change directory (CD) option along with the macro variable (&a) that contains the location of the work folder obtained in step one. The make directory option (MKDIR) is then used to create the directory (subfolder). In the final step the library name is created using the macro variable created in step two. The option NOXWAIT should also be specified as it instructs SAS to automatically close the prompt window after the execution of a specified command, eliminating the need to use the EXIT command when using DOS functions.

```
options noxwait;

/*1. Get the address of work folder */

%let a = %sysfunc(getoption(work));
%put &a;
%global addr;

/*2. Create macro variable to be used in the libname statement */

data _null_;
  addr="&a"||"\temp";
  call symput('addr',addr);
run;

/*3. Use the command prompt (DOS) commands (change directory and make
directory) to create the subfolder */

x "cd &a";
x "mkdir temp" ;

/*4. Create the libname */

libname raw "&addr";
```

ODS TRICKS

Some of the ODS tricks that were used in implementing the patient profile are as follows.

1. When using ODS in combination with PROC REPORT there is no direct approach to create subheaders. To create subheaders the subtitle is included as part of the column statement. In the example below the subtitle is "DEMOGRAPHIC INFORMATION". The ODS ESCAPE CHARACTER used is "~".

```
column ("~R'\q1' ~R' \b' ~R' \u1' DEMOGRAPHIC INFORMATION ~R' \u10' ~R' \b0'  
      ( age sex race ethnic ));
```

DEMOGRAPHIC INFORMATION

Age (years)	Gender	Race	Ethnicity
46	F	WHITE	NOT HISPANIC OR LATINO

2. In a patient profile algorithm each report is printed one after the other in a loop. Sometimes the reporting is such that two subtitles may be involved. The first subtitle is created using the COMPUTE AFTER block in the PROC REPORT definition and the second subtitle is implemented as described in the above step.

```
COMPUTE AFTER;  
LINE ' ';  
LINE "~R'\b 'SUPPLEMENTARY DATA~R'\b0' ";  
ENDCOMP;
```

3. In the template definition, each of the rows in the patient profile output can be output in a tabular form or the gridlines can be avoided using the RULES=NONE option.

CONCLUSION

A robust and very structured methodology to implement a patient profile has been shown. As discussed, there is still a need for the creation of simple patient profiles to meet operational objectives when conducting clinical trials, creating the submission documents, and making patient safety decisions. The techniques the authors of this paper are proposing are for operational uses, not submission uses, of patient profiles as in aiding narrative writing, DSMB, and internal data review meetings. Numerous features of ODS RTF in conjunction with PROC REPORT and PROC PRINT have been used to simplify the presentation of the data.

REFERENCES

FDA Guidance for Industry (1997), Archiving Submissions in Electronic Format - NDAs

Light S, Gilbert P, Genereux G, A Novel Approach to Developing a Patient Profile Reporting Application, SUGI 26, p042-26

Use ODS Generating Patient Profiles, Haibin Shu, Young Zhuge. PharmaSUG 2005

Validation Checks Performed by WebSDM™ on SDTM version 3.1.1 Datasets, V1.5:
<http://www.phaseforward.com/products/safety/documents/ValidationChecksPerformedbyWebSDMtm.Q107.pdf>

Guidance for Industry - Providing Regulatory Submissions in Electronic Format – NDAs, U.S. Department of Health and Human Services, Food and Drug Administration, Center for Drug Evaluation and Research (CDER), IT 3, January, 1999.

ACKNOWLEDGMENTS

The authors would like to thank Octagon Research Solutions for their encouragement and support. The authors would also like to thank Allan Glaser and Brian Shilling for critiquing the paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Terek Peterson, MBA
Director, Clinical Programming
Octagon Research Solutions
585 E. Swedesford Road, Suite 200
Wayne, PA 19087
(w) (610) 535-6500 x5658

Sanjiv Ramalingam, MS
Specialist, Clinical Programming
Octagon Research Solutions
585 E. Swedesford Road, Suite 200
Wayne, PA 19087
sramalingam@octagonresearch.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.